Esame del corso di Verifica di Programmi Prolog Reali

Soluzione dell’esercizio 6 della “Prolog Competition” dell’ILPS’94

Marco Comini e Enea Zaffanella

Dipartimento di Informatica Corso Italia, 40 { 56125 Pisa f comini; zaffanel g@di:unipi:it

14 marzo 1996

Sommario

Mostriamo un programma per risolvere l’esercizio 6 della \Prolog Competition" dell’ILPS’94 e ne mostriamo la correttezza parziale e la terminazione.

1 Il Programma

Assunzioni Preliminari. Per semplificare la notazione rispetto al testo rappresentiamo ogni casella (x, y) rimossa con il fatto goto(x, y, b) anziché con l’assenza di un fatto tipo goto(x, y, m) (con m ∈ { l, r, u, d, f }).

Inoltre indeboliamo la specifica del testo dell’esercizio eliminando alcune ipotesi. Siccome non ci interessa garantire l’unicità del percorso corretto, ammettiamo che due caselle adiacenti sul cammino siano bianche e che eventualmente più di due caselle adiacenti ad una casella bianca appartengano ad un percorso corretto. Assumiamo che siano invece soddisfatte tutte le altre ipotesi del testo.

Il programma che risolve questo problema è il seguente.

path(Path) :-
    path([(1,1)], Path).

path([(C,R)|SoFar], Path) :-
    goto(C, R, f),
    reverse([(C,R)|SoFar], Path).
path([(C,R)|SoFar], Path) :-
    size(Size),
    getMove((C,R), Move),
    nextPos((C,R), Move, Size, (NextC,NextR)),
    noCycle([(C,R)|SoFar], (NextC,NextR)),
    path([(NextC,NextR),(C,R)|SoFar], Path).

getMove((C,R), Move) :-
    goto(C, R, Move).
getMove((C,R), Move) :-
    goto(C, R, b),
    guessMove(Move).

guessMove(l).
guessMove(d).
guessMove(r).
guessMove(u).

nextPos((C,R), u, Size, (C,NextR)) :-
    R < Size,
    NextR is R + 1.

nextPos((C,R), d, Size, (C,NextR)) :-
    R > 1,
    NextR is R - 1.

nextPos((C,R), r, Size, (NextC,R)) :-
    C < Size,
    NextC is C + 1.

nextPos((C,R), l, Size, (NextC,R)) :-
    C > 1,
    NextC is C - 1.

noCycle([], Next).
noCycle([X|Rest], Next) :-
    Next \== X,
    noCycle(Rest, Next).

reverse(Xs, Ys) :-
    reverse(Xs, [], Ys).

reverse([], Acc, Acc).
reverse([First|Rest], Acc, Ys) :-
    reverse(Rest, [First|Acc], Ys).

A questo programma vanno aggiunte le clausole che definiscono la particolare mappa, e cioè un unico fatto size(n) che definisce la misura della mappa e un insieme di fatti del tipo goto(x, y, m) (m ∈ { l, r, u, d, f, b }).

Il nostro programma fornisce (in backtracking) tutti i percorsi corretti contenuti nella mappa. In particolare, nel caso in cui la mappa soddisfi anche le due ipotesi rimosse dalla specifica originale, si ottiene l’unico percorso corretto.

Esempio 1.1 Aggiungendo al programma sui seguenti fatti

size(5).

goto(1, 1, b). goto(2, 1, d). goto(3, 1, d).
goto(4, 1, f). goto(5, 1, l).

goto(1, 2, u). goto(2, 2, r). goto(3, 2, b).
goto(4, 2, l). goto(5, 2, b).

goto(1, 3, b). goto(2, 3, d). goto(3, 3, u).
goto(4, 3, d). goto(5, 3, d).

goto(1, 4, d). goto(2, 4, u). goto(3, 4, l).
goto(4, 4, d). goto(5, 4, b).

goto(1, 5, d). goto(2, 5, b). goto(3, 5, r).
goto(4, 5, b). goto(5, 5, d).

che corrispondono alla mappa

=⇒

↓ → ↓ ↓ ↑ ← ↓ ↓ ↑ ↓ ↓ ↑ → ← ↓ ↓ • ← otteniamo come output

| ?- path(X).

X = [(1,1), (1,2), (1,3), (2,3), (2,2), (3,2), (3,3),
(3,4), (2,4), (2,5), (3,5), (4,5), (5,5), (5,4), (5,3),
(5,2), (5,1), (4,1)] ? ;

no

che corrisponde all’unico cammino corretto

=⇒

→ → → ↓ ↑ ← ↓ → ↓ ↑ ↓ ↑ → ↑ ↓ ↑ • ←

Esempio 1.2 Modificando l’Esempio 1.1 con i fatti goto(1, 2, b) e goto(3,

4, b) al posto di goto(1, 2, u) e goto(3, 4, l) che corrispondono alla

mappa

=⇒

↓ → ↓ ↓ ↑ · ↓ ↓ ↑ ↓ ↓ · → ← ↓ ↓ • ← otteniamo invece come output

| ?- path(X).

X = [(1,1), (1,2), (2,2), (3,2), (3,3), (3,4), (2,4),
(2,5), (3,5), (4,5), (5,5), (5,4), (5, 3), (5,2), (5,1),
(4,1)] ? ;

X = [(1,1), (1,2), (2,2), (3,2), (3,3), (3,4), (3,5),
(4,5), (5,5), (5,4), (5,3), (5,2), (5, 1), (4,1)] ? ;

X = [(1,1), (1,2), (1,3), (2,3), (2,2), (3,2), (3,3),
(3,4), (2,4), (2,5), (3,5), (4,5), (5, 5), (5,4), (5,3),
(5,2), (5,1), (4,1)] ? ;

X = [(1,1), (1,2), (1,3), (2,3), (2,2), (3,2), (3,3),
(3,4), (3,5), (4,5), (5,5), (5,4), (5, 3), (5,2), (5,1),
(4,1)] ? ;

no

2 Analisi del programma

2.1 Modi Input/Output, Tipi, Occur-Check Freeness e Errori a Run-time

Definiamo i tipi P os = { (x, y) | x, y ∈ Int } M ove = { u, d, l, r, f, b } List = liste di elementi qualsiasi P osList = liste di elementi di P os Gae = Ground arithmetic expressions

Date le seguenti definizioni di modi e di tipo per i predicati del programma path(Out:PosList) path(In:PosList, Out:PosList) getMove(In:Pos, Out:Move) guessMove(Out:Move) nextPos(In:Pos, In:Move, In:Int, Out:Pos) noCycle(In:PosList, In:Pos) reverse(In:List, Out:List) reverse(In:List, In:List, Out:List) size(Out:Int) goto(Out:Int, Out:Int, Out:Move) \==(In:Int, In:Int) >(In:Int, In:Int) <(In:Int, In:Int) is(Out:Int, In:Gae)

si dimostra facilmente che il programma (rispetto al goal path(P ath)) è ben modato e ben tipato.

Proposizione 2.1 Programma e goal sono “occur-check free”.

Dimostrazione. Il programma e il goal sono ben modati. Inoltre tutte le teste di clausole di P sono “output linear”. Possiamo quindi applicare il Corollario 4.5 di [1] da cui discende la tesi.

Proposizione 2.2 Programma e goal sono liberi da errori run-time.

Dimostrazione. Le chiamate a predicati built-in che vengono fatte dal programma nel risolvere il goal rispettano le modalità e i tipi ammessi.

2.2 Correttezza Parziale e Terminazione

Introduciamo ora un po’ di terminologia e notazioni che ci serviranno per dimostrare la correttezza parziale del programma. Con lettere minuscole denotiamo variabili a valori nell’Universo di Herbrand. Con λ denotiamo la lista vuota, mentre se x è una lista, con e · x denotiamo la lista ottenuta aggiungendo e in testa ad x. Date due liste x, y con concat(x, y) denotiamo la lista che si ottiene concatenando x e y. Se x = e_1 . . . e_n è una lista, con reverse(x) denotiamo la lista e_n . . . e_1, |x| denota la lunghezza n di x, mentre prefix(x) denota l’insieme { e_1 . . . e_i | i ≤ n }. Inoltre diciamo che e ∈ x se e occorre in x.

Data E ∈ Gae, con eval(E) denotiamo il termine in Int che corrisponde alla valutazione di E.

D’ora in poi N denoterà la dimensione della mappa definita dal fatto size(N) ∈ P . Definizione 2.3 Una coppia (c, r) è ammissibile se 1 ≤ c, r ≤ N . L’insieme delle coppie ammissibili è CA ⊆ P os.

Assumiamo inoltre che sia definita una funzione goto : CA −→ M ove che ad ogni coppia associa la corrispondente mossa presente sulla scacchiera.

Definiamo una relazione^{T A}−−→ per mettere in corrispondenza coppie (ammissibili) del percorso sotto esame.^{T A}−−→ è data dall’unione delle relazioni^{T A}−−→_u ,

T A−−→_d ,^{T A}−−→_r e^{T A}−−→_l definite come segue. (c, r) ∈ CA, (goto(c, r, u) ∈ P ∨ goto(c, r, b) ∈ P ), (c, r + 1) ∈ CA (c, r)^{T A}−−→_u (c, r + 1)

(c, r) ∈ CA, (goto(c, r, d) ∈ P ∨ goto(c, r, b) ∈ P ), (c, r − 1) ∈ CA (c, r)^{T A}−−→_d (c, r − 1)

(c, r) ∈ CA, (goto(c, r, r) ∈ P ∨ goto(c, r, b) ∈ P ), (c + 1, r) ∈ CA (c, r)^{T A}−−→_r (c + 1, r)

(c, r) ∈ CA, (goto(c, r, l) ∈ P ∨ goto(c, r, b) ∈ P ), (c − 1, r) ∈ CA (c, r)^{T A}−−→_l (c − 1, r) Definizione 2.4 Un percorso parziale è una lista σ_1 · · · σ_n senza ripetizioni di coppie ammissibili tale che σ_1 = (1, 1) e ∀i ∈ { 1, . . . , n − 1 } σ_i^{T A}−−→ σ_{i+1}. P P D ⊆ P osList è l’insieme dei percorsi parziali. Inoltre P P R = { x | reverse(x) ∈ P P D } Un percorso parziale è detto percorso finale se, posto che (c, r) sia il suo ultimo elemento, abbiamo goto(c, r) = f. L’insieme dei percorsi finali è denotato P F . Lemma 2.5 Se x ∈ P P R allora |x| ≤ N^2.

Dimostrazione. x è una lista di elementi di CA senza ripetizioni e card(CA) = N^2.

Per definizione, tutte le clausole unitarie sono ricorrenti. Inoltre, gli unici predicati ricorsivi sono path/2, noCycle/2, reverse/3 e non si hanno predicati mutuamente ricorsivi. Nel seguito ometteremo la dimostrazione di terminazione quando risulti essere banale.

Definiamo i seguenti level mapping: |path(x)| = 2N^2 |path(x, y)| = 2N^2 − |x| |noCycle(x, p)| = |x| |reverse(x, y)| = |x| |reverse(x, y, z)| = |x| Gli atomi getM ove(x, m), guessM ove(m), nextP os(x, m, y), goto(c, r, m) e size(n) che i built-in (is, >, <, \==) sono invece mappati a zero.

Definiamo ora P re e P ost per i predicati del programma. path(T ) { P re = { path(x) } P ost = { path(x) | x ∈ P F } path(T_1, T_2) { P re = { path(t_1, t_2) | t_1 ∈ P P R } P ost = { path(t_1, t_2) | t_1 ∈ P P R, t_2 ∈ P F, reverse(t_1) ∈ prefix(t_2) } getM ove(C, M ) P re = { getM ove(p, m) } P ost = { getM ove((c, r), m) | (c, r) ∈ CA, m = goto(c, r) ∨ (goto(c, r) = b ∧ m ∈ { u, d, l, r }) }

guessM ove(M ) { P re = { guessM ove(m) } P ost = { guessM ove(m) | m ∈ { u, d, l, r } } nextP os(C_1, M, Size, C_2) P re = { nextP os(c_1, m, N, c_2) | c_1 ∈ CA } P ost = { nextP os(c_1, m, N, c_2) | c_1, c_2 ∈ CA, c_1^{T A}−−→_m c_2 }

noCycle(L, E) { P re = { noCycle(x, e) } P ost = { noCycle(l, e) | l ∈ List, e ∉ l } reverse(Xs, Y s) { P re = { reverse(x, y) } P ost = { reverse(x, y) | x, y ∈ List, y = reverse(x) } reverse(Xs, Acc, Y s) { P re = { reverse(x, y, z) | y ∈ List } P ost = { reverse(x, y, z) | x, y, z ∈ List, z = concat(reverse(x), y) }

size(N ) { P re = { size(n) } P ost = { size(N ) } goto(C, R, M ) { P re = { goto(c, r, m) } P ost = { goto(c, r, m) | (c, r) ∈ CA, m ∈ M ove } > (X, Y ) { P re = { > (x, y) } P ost = { > (x, y) | x, y ∈ Int, x > y } < (X, Y ) { P re = { < (x, y) } P ost = { < (x, y) | x, y ∈ Int, x < y } is(X, Y ) { P re = { is(x, y) } P ost = { is(x, y) | x ∈ Int, y ∈ Gae, x = eval(y) } \ == (X, Y ) { P re = { \ == (x, y) } P ost = { \ == (x, y) | x ≠ y } Teorema 2.6 Il programma è parzialmente corretto e termina^1.

1Bisogna sempre avere almeno un teorema nella vita.

Dimostrazione. Per goto/3, guessMove/1, size/1 la dimostrazione è immediata. I built-in sono parzialmente corretti per definizione. Forniamo ora i “proof outlines” per le restanti clausole [2]. { path(x) } path(P ath) : − ⟨t_0 = 2N^2⟩ { path((1, 1) · λ, x) } %%(1, 1) · λ ∈ P P R%% path([(1, 1)], P ath). ⟨t_1 = 2N^2 − |[(1, 1)]| = 2N^2 − 1⟩ { path((1, 1) · λ, x) | x ∈ P F } { path(x) | x ∈ P F } { path((c, r) · x, y) | (c, r) · x ∈ P P R } path([(C, R)|SoF ar], P ath) : − ⟨t_0 = 2N^2 − |[(C, R)|SoF ar]| ≥ N^2⟩ { goto(c, r, f) } goto(C, R, f), ⟨t_1 = 0⟩ { goto(c, r, f) | (c, r) ∈ CA, goto(c, r) = f } { reverse((c, r) · x, y) } reverse([(C, R)|SoF ar], P ath). ⟨t_2 = |[(C, R)|SoF ar]| ≤ N^2 ≤ t_0 } { reverse((c, r) · x, y) | y = reverse((c, r) · x) } { path((c, r) · x, y) | (c, r) · x ∈ P P R, y ∈ P F, y = reverse((c, r) · x) }

{ path((c, r) · x, y) | (c, r) · x ∈ P P R } path([(C, R)|SoF ar], P ath) : − ⟨t_0 = 2N^2 − |[(C, R)|SoF ar]| ≥ N^2⟩ { size(n) }, size(Size), ⟨t_1 = 0⟩ { size(N ) }, { getM ove((c, r), m) } getM ove((C, R), M ove), ⟨t_2 = 0⟩ { getM ove((c, r), m) | (c, r) ∈ CA, m = goto(c, r)∨ (goto(c, r) = b ∧ m ∈ { u, d, l, r }) } { nextP os((c, r), m, N, (c^0, r^0)) | (c, r) ∈ CA } nextP os((C, R), M ove, Size, (N extC, N extR)), ⟨t_3 = 0⟩ { nextP os((c, r), m, N, (c^0, r^0)) | (c, r), (c^0, r^0) ∈ CA, (c, r)^{T A}−−→_m (c^0, r^0) } { noCycle((c, r) · x, (c^0, r^0)) } noCycle([(C, R)|SoF ar], (N extC, N extR)), ⟨t_4 = |[(C, R)|SoF ar]| ≤ N^2 ≤ t_0⟩ { noCycle((c, r) · x, (c^0, r^0)) | (c, r) · x ∈ List, (c^0, r^0) ∉ (c, r) · x } { path((c^0, r^0) · (c, r) · x, y) | (c^0, r^0) · (c, r) · x ∈ P P R } path([(N extC, N extR), (C, R)|SoF ar], P ath). ⟨t_5 = 2N^2 − (|SoF ar| + 2) = t_0 − 1⟩ { path((c^0, r^0) · (c, r) · x, y) | (c^0, r^0) · (c, r) · x ∈ P P R, y ∈ P F , reverse((c^0, r^0) · (c, r) · x) ∈ prefix(y) } { path((c, r) · x, y) | (c, r) · x ∈ P P R, y ∈ P F , reverse((c, r) · x) ∈ prefix(y) }

{ getM ove((c, r), m) } getM ove((C, R), M ove) : − { goto(c, r, m) } goto(C, R, M ove). { goto(c, r, m) | (c, r) ∈ CA, m = goto(c, r) } { getM ove((c, r), m) | (c, r) ∈ CA, m = goto(c, r) }

{ getM ove((c, r), m) } getM ove((C, R), M ove) : − { goto(c, r, b) } goto(C, R, b), { goto(c, r, b) | (c, r) ∈ CA, goto(c, r) = b } { guessM ove(m) } guessM ove(M ove). { guessM ove(m) | m ∈ { u, d, l, r } } { getM ove((c, r), m) | (c, r) ∈ CA, goto(c, r) = b, m ∈ { u, d, l, r } } { nextP os((c, r), u, N, (c, r^0)) | (c, r) ∈ CA } nextP os((C, R), u, Size, (C, N extR)) : − { < (r, N ) } R < Size, { < (r, N ) | r < N } { is(r^0, +(r, 1)) } N extR is R + 1. { is(r^0, +(r, 1)) | r^0 ∈ Int, +(r, 1) ∈ Gae, r^0 = eval(+(r, 1)) = r + 1 } { nextP os((c, r), u, N, (c, r^0)) | (c, r), (c, r^0) ∈ CA, (c, r)^{T A}−−→_u (c, r^0) }

Per le altre tre clausole con d, l, r si procede analogamente. { noCycle(λ, p) } noCycle([], N ext). { noCycle(λ, p) } { noCycle(e · x, p) } noCycle([X|Rest], N ext) : − ⟨t_0 = |Rest| + 1⟩ { \ == (e, p) } N ext\ == X, ⟨t_1 = 0⟩ { \ == (e, p) | e ≠ p } { noCycle(x, p) } noCycle(Rest, N ext). ⟨t_2 = |Rest| = t_0 − 1 } { noCycle(x, p) | p ∉ x } { noCycle(e · x, p) | p ∉ e · x }

{ reverse(x, y) } reverse(Xs, Y s) : − ⟨t_0 = |Xs|⟩ { reverse(x, a, y) | a ∈ List } reverse(Xs, [], Y s). ⟨t_1 = |Xs| = t_0⟩ { reverse(x, λ, y) | x, y ∈ List, y = concat(reverse(x), λ) } { reverse(x, y) | x, y ∈ List, y = reverse(x) } { reverse(λ, z, y) | z ∈ List } reverse([], Acc, Acc). { reverse(λ, z, y) | y, z ∈ List, y = concat(reverse(λ), z) = z } { reverse(e · x, z, y) | z ∈ List } reverse([F irst|Rest], Acc, Y s) : − ⟨t_0 = |Rest| + 1⟩ { reverse(x, e · z, y) | e · z ∈ List } reverse(Rest, [F irst|Acc], Y s). ⟨t_1 = |Rest| = t_0 − 1⟩ { reverse(x, e · z, y) | x, y, z ∈ List, y = concat(reverse(x), e · z) } { reverse(e · x, z, y) | e · x, y, z ∈ List, y = concat(reverse(e · x), z) }

Riferimenti bibliografici

[1] K. R. Apt and A. Pellegrini. On the occur-check free prolog programs.

Technical Report CS-R9238, CWI, October 1992.

[2] D. Pedreschi and S. Ruggieri. A case study in logic program verification:

the vanilla metainterpreter. In Proceedings of the GULP-PRODE’95 Joint Conference on Declarative Programming, 1995.